
Abstract. Clinical practice guidelines (CPGs) were originally designed
to help with evidence-based management of a single disease and such sin-
gle disease focus has impacted research on CPG computerization. This
computerization is mostly concerned with supporting different represen-
tation formats and identifying potential inconsistencies in the definitions
of CPGs. However, one of the biggest challenges facing physicians is
the application of multiple CPGs to comorbid patients. While various
research initiatives propose ways of mitigating adverse interactions in
concurrently applied CPGs, there are no attempts to develop a general-
ized framework for mitigation that captures generic characteristics of the
problem, while handling nuances such as precedence relationships. In this
paper we present our research towards developing a mitigation framework
that relies on a first-order logic-based representation and related theorem
proving and model finding techniques. The application of the proposed
framework is illustrated with a simple clinical example.

1 Introduction

A clinical practice guideline (CPG) codifies the evidence-based best practice in
prescribing the most appropriate disease-specific therapy to patients, subject
to available patient data and possible diagnoses [16]. Since the scope of each
guideline is limited to a single disease, the evidence-based management of a
comorbid patient according to the recommendations concurrently coming from
multiple CPGs is difficult and can result in inconsistent and potentially harmful
therapies. Often the derivation of a combined therapy directly from the guidelines
(even for properly diagnosed comorbid conditions) is incorrect due to adverse
interactions between the treatments associated with individual therapies. These
interactions manifest directly as contradictory recommendations (e.g., use of
steroids is recommended by one CPG and prohibited by the other), or they may
correspond to drug-drug or drug-disease adverse interactions resulting in actions
that cannot be taken concurrently.

As a matter of fact, concurrent application of two or more CPGs is chal-
lenging – it requires designing a sophisticated mechanism for identifying and

eliminating potential redundancy in the tests or procedures, identifying contra-
dictions (direct adverse interactions), and for managing discordance (indirect,
drug-drug or drug-disease interactions) [17]. As such, it is believed that exe-
cuting multiple CPGs concurrently requires a new, “combinatorial, logical, or
semantic” methodological approach [2].

Our previous research [7,8,19] proposes such an approach by introducing
and formally defining logical models of CPGs and developing a mitigation algo-
rithm that operates on these models. The algorithm relies on secondary clinical
knowledge (i.e., knowledge that goes beyond the primary knowledge encoded in
CPGs and that comes from domain experts, textbooks, or repositories of clinical
evidence) that is encoded as interaction and revision operators. The operators
characterize adverse interactions associated with the concurrent application of
CPGs and describe revisions to logical models required to address these interac-
tions. The algorithm employs the constraint logic programming (CLP) paradigm
to efficiently solve the logical models, where a solution represents a combined
therapy free of adverse interactions.

In the research described here, we move further towards developing a general
framework for mitigation by enriching the representation of CPGs using first-
order logic (FOL) theories and relying on theorem proving and model finding
techniques to process these theories. This expansion is dictated by the following
limitations of our previous research:

– Restricted expressive power of the CLP-based approach that does not allow for
explicit representation of properties of objects (e.g., a dosage associated with
a specific CPG action) and relationships between objects (e.g., precedence
between CPG actions),

– Limited interpretability of solutions returned by CLP solvers and consequently
the need to assign real-world semantics to truth-value assignment of the propo-
sitional symbols in the CLP-based model.

FOL significantly improves the expressiveness of our approach by introducing
predicates to represent properties and relationships in the domain (in fact, rela-
tionships are only first-order definable). Moreover, predicates impose semantics
on solutions, facilitating their interpretation from a clinical perspective.

This paper is organized as follows. We start with a brief review of related
work. Then, we present the foundations of FOL, theorem proving and model find-
ing that are relevant to our research. Next, we describe the proposed framework –
we start with the underlying FOL theories and then present an overview of the
mitigation process. We proceed with a simple clinical example that illustrates
the application of the framework. Finally, we provide conclusions and directions
for our future research.

2 Related Work

Peleg in her recent methodological review [12] divided the research on computer-
interpretable CPGs into eight themes: (1) modeling, (2) acquisition and

specification, (3) integration in combination with electronic patient record, (4) val-
idation and verification, (5) execution, (6) exception handling, (7) maintenance,
and finally (8) sharing. According to such categorization, our research discussed
here belongs to the validation and verification theme. This theme is further sub-
divided into three problems: (1) checking for internal consistency and existence
of anomalies, (2) checking for existence of desired properties and (3) checking for
the inconsistencies between multiple CPGs applied to a comorbid patient.

The first problem from the above list was addressed for example in [1], where
the authors proposed a knowledge-based detection method for checking the con-
sistency of a CPG represented in ASBRU. The second problem was described in
[14], where model checking was applied for authoring and verification of CPGs
represented in UML. Moreover, in [18] theorem proving techniques were used
to check whether a guideline for managing jaundice in newborns complies with
certain properties. Finally, [13] described a comprehensive framework employing
ontological domain knowledge and abductive reasoning to evaluate the complete-
ness and appropriateness of a CPG, and to assess the compliance of physician’s
actions with this CPG.

The research related to the last verification problem is still in its relatively
early stages despite its clinical importance. Proposed solutions vary from manual
interventions, where human experts verify and combine multiple CPGs using a
specialized editing tool [15], through semi-automatic approaches, where experts
resolve automatically discovered conflicts [3], to fully automatic techniques [4,5].
In [4] the authors proposed an approach that operates on ontological models of
CPGs and applies ontology merging techniques to combine these models so that
medical, work-flow, institutional and temporal constraints are satisfied. A differ-
ent approach was described in [5], where individual CPGs are merged according
to the combination rules that capture possible drug-drug interactions and pre-
scribe ways of avoiding them.

3 Background

3.1 Foundations of FOL

The formal language of FOL relies on logical and non-logical symbols. The logical
symbols (connectives, quantifiers, variables) are those that have a fixed mean-
ing in a language. The non-logical symbols are those that have an application-
dependent meaning (e.g., symbols needed to represent a CPG in FOL) and
they are further categorized into function symbols and predicate symbols. Each
non-logical symbol has an arity, indicating how many arguments it requires.
A function symbol with arity 0 is called a constant and a predicate symbol with
arity 0 is called a propositional symbol.

FOL allows for two types of syntactic expressions: terms (made of variables,
constants and functions) and formulas (composed of terms, predicates and con-
nectives). Formulas with variables bounded by quantifiers and formulas without
variables (i.e., grounded formulas) are called sentences. A FOL theory D is a
collection of sentences.

An interpretation I (sometimes called a structure) in FOL is defined as triple:

I = 〈Idomain, Ipredicate, Ifunction〉 , (1)

where

– Idomain is any nonempty set of objects under consideration called the domain
of the interpretation,

– Ipredicate is a set of interpretation mappings over Idomain,
– Ifunction is a set of functions over Ifunction.

Mappings from Ipredicate assign meaning to the predicate symbols as follows:
for every predicate symbol P of arity n, I[P ] ∈ Ipredicate is an n-ary relation
over Idomain, that is I[P ] ⊆ Idomain × . . . × Idomain.

Mappings from Ifunction assign meaning to the function symbols as follows:
for every function symbol F of arity n, I[F ] ∈ Ifunction is an n-ary function
over Idomain, that is I[F ] ∈ [Idomain × . . . × Idomain → Idomain].

Given an interpretation I, we can check which sentences of a FOL theory D
are true and which are false according to this interpretation. If a sentence φ ∈ D
is true given I, then we write it formally as I |=m φ. Moreover, if I satisfies all
sentences in D, then it is called a model for theory D and formally it is denoted
as I |=m D.

3.2 Theorem Proving and Model Finding

There are three fundamental questions that are associated with FOL theories:

1. Is a given theory consistent?
2. What is a model for a consistent theory?
3. What are logical consequences (implications) of a consistent theory?

A FOL theory D is consistent (or satisfiable), iff there exists at least one
model of this theory. The question on the consistency of D can be answered using
theorem proving [11] that employs automatic reasoning (the resolution method)
to construct a proof for D. However, theorem proving techniques provide only
a binary answer to the consistency question and no model is directly returned,
even if it exists (i.e., when the answer is positive). In order to answer the question
about a model for a consistent theory, one needs to use model finding techniques
that can be considered as a special case of solving the constraint satisfaction
problem [20], where possible interpretations are generated until a model is found.

The question about logical consequences is translated into checking if a FOL
theory D entails sentence φ (or φ is a logical consequence of D). Formally, we say
D entails φ, written as D |= φ, iff, for every interpretation I such that I |=m D,
we have I |=m φ. In other words, we say D entails φ (or φ can be deduced from
D), if φ is satisfied by all models for D.

The entailment D |= φ can be translated into checking whether a new theory
D ∪ {¬φ} is not consistent. This means that theorem proving techniques can
equivalently be used to check for logical entailments of a theory D.

Table 1. Defined predicates

Predicate Description

node(x) x is a node in AG

action(x) x is an action node in AG

decision(x) x is a decision node in AG

executed(x) Action node x is executed

value(x, v) Value v is associated with decision node x

dosage(x, n) Action node x is characterized by medication dosage n

directPrec(x, y) Node x directly precedes node y (there is an edge from x to y)

prec(x, y) Node x precedes node y (there is a path from x to y)

disease(d) d is a disease to be managed

diagnosed(d) The patient has been diagnosed with disease d

4 Methodology

Using FOL in a framework for the mitigation of concurrently applied CPGs relies
on four key components that are listed below and described in the following
sections:

1. A vocabulary used to construct the FOL theory describing a particular miti-
gation problem (further referred to as to combined mitigation theory),

2. A combined mitigation theory composed of individual theories that describe
various aspects of the mitigation problem,

3. A set of operators that encode the secondary knowledge needed to identify
and address adverse interactions associated with the combined mitigation
theory,

4. A mitigation algorithm that controls the application of operators to the com-
bined mitigation theory.

4.1 Vocabulary

Following our previous work, we assume a CPG is represented as an actionable
graph (AG) [19]. An AG is a directed graph composed of three types of nodes
context, action, and decision, and arcs that represent transitions between nodes.
A context node defines an entry point and indicates the disease associated with
the CPG, an action node indicates a clinical action that needs to be executed,
and a decision node indicates a selection from several alternative choices and
allows for conditional branching.

The vocabulary of our FOL-based approach is composed of constants (denoted
with upper case letters), variables (denoted with lower case letters) and predicates.
The predicates used in the mitigation problem are listed in Table 1. We note there
is no predicate corresponding to a context node, as information embedded in this
node is provided by the predicate disease(d).

4.2 Combined Mitigation Theory

We use the vocabulary to construct a combined mitigation theory. Formally, this
combined theory Dcomb is defined as a triple:

Dcomb = 〈Dcommon,Dcpg,Dpi〉 , (2)

where:

– Dcommon is a theory that axiomatizes the universal characteristics of CPGs
as part of the FOL representation. It is the common (shared and reusable)
component of all mitigation theories and it contains the following axioms (for
brevity we limit the list to the most relevant ones):
• ∀x, y directPrec(x, y) ⇒ prec(x, y) – association between precedence and

direct precedence,
• ∀x, y, z prec(x, y) ∧ prec(y, z) ⇒ prec(x, z) – transitivity of precedence,
• ∀x, y, prec(x, y) ∧ prec(y, x) ⇒ x = y – antisymmetry of precedence,
• ∀x node(x) ⇒ (action(x) ∧ ¬decision(x)) ∨ (¬action(x) ∧ decision(x)) –

ensures that a node cannot be simultaneously an action and decision node,
• ∀x, n dosage(x, n) ⇒ action(x) – ensures that only an action node can be

characterized with medication dosage,
• ∀x, v value(x, v) ⇒ decision(x) – ensures that only a decision node can be

characterized by a value,
• ∀d diagnosed(d) ⇒ disease(d) – ensures that the diagnosed disease is the

same as the disease to be managed.
– Dcpg is a union of theories, each theory representing a single AG (and thus

the underlying CPG) that are being applied to a comorbid patient:

Dcpg = Dd1
cpg ∪ Dd2

cpg ∪ . . . ∪ Ddk
cpg, (3)

where Ddi
cpg is the theory that describes the AG associated with disease di by

enlisting all nodes and paths, giving information about precedence between
nodes and providing information on dosages associated with selected action
nodes. Because of axioms in Dcommon it is sufficient to define only direct
precedence between nodes (directPrec predicate) – precedence between nodes
represented with the prec predicate is derived automatically,

– Dpi is the theory that describes available patient information. It contains sen-
tences representing patient data, including results of tests and examinations,
and indicating already prescribed therapies and procedures.

4.3 Interaction and Revision Operators

Interaction and revision operators were introduced in our previous research [19].
Here we reformulate them to account for the FOL-based representation and to
enhance their capabilities (e.g., a revision operator may specify multiple opera-
tions – details provided below). An interaction operator IOk encodes knowledge



Using First-Order Logic to Represent Clinical Practice Guidelines 51

about indirect adverse interactions (usually drug-drug or drug-disease) and for-
mally it is defined as:

IOk =
〈
αk

〉
, (4)

where αk is a sentence (constructed with the vocabulary described in Sect. 4.1)
describing a specific indirect interaction. Checking whether IOk is applicable to
Dcomb (or in other words, if the interaction represented by IOk occurs in Dcomb)
is an entailment problem Dcomb |= αk.

A revision operator encodes knowledge about the revisions that need to be
introduced to the theory Dcpg in order to address encountered interactions (both
direct and indirect). In layman terms, it describes changes that need to be
introduced to concurrently applied CPGs. Formally, a revision operator ROk is
defined as:

ROk =
〈
βk, Opk

〉
, (5)

where βk is a logical sentence that defines the applicability of the operator to the
theory Dcpg, and Opk describes the revisions introduced by ROk. In particular,
Opk is a set of n pairs of formulas

〈
φk
i , ψ

k
i

〉
(i = 1 . . . n) that define single opera-

tions within the operator. As already stated, these operations are applied only to
Dcpg, so other components of Dcomb are protected from unwanted revisions. For
example, Dpi is never modified thus patient information is never inadvertently
changed. The pairs of formulas are interpreted as follows (∅ indicates an empty
formula):

–
〈
φk
i , ∅

〉
means that φk

i is removed from any sentence in Dcpg where it appears,
–

〈∅, ψk
i

〉
means that ψk

i is added as a new sentence to Dcpg,
–

〈
φk
i , ψ

k
i

〉
means that φk

i is replaced by ψk
i in any sentence in Dcpg where it

appears.

It is possible to use unbounded variables in φk
i and ψk

i and these are inter-
preted as “wildcards” that are bound to a constant specific to a patient encounter
when revisions are being introduced. For example, one can define an operation
that increases the dosage of a medication by a given amount. Moreover, checking
the applicability of ROk to Dcomb is analogous to checking the applicability of
IOk and translates into the entailment problem Dcomb |= βk. In case of direct
interactions this entailment problem is simplified – details are given in the next
section.

4.4 Mitigation Algorithm

The algorithm consists of two phases and it is outlined in Fig. 1. The first phase
involves mitigating direct adverse interactions. Their identification translates
into checking the consistency of the Dcomb theory (note that in order to check
for consistency and entailment we need to create a temporary theory that is
a union of all three components in Dcomb). If the theory is consistent, then it
indicates there are no direct interactions and the algorithm passes to the second
phase. Otherwise, the theory Dcomb (specifically its Dcpg component) needs to
be revised using applicable revision operators.

Fig. 1. Outline of the mitigation algorithm.

Since Dcomb is inconsistent, entailment cannot be used to find applicable revi-
sion operators, as entailment problems can only be formulated over a consistent
theory. Instead, we use the following procedure. First, we identify actions shared
across individual theories (i.e., theories representing single CPGs) in Dcomb.
Then, for each shared action xs we check whether execution of this action and
its negation are entailed by individual theories (i.e., Ddi

cpg |= executed(xs) and
Ddj

cpg |= ¬executed(xs)). Such entailments indicate inconsistency caused by xs.
Finally, we identify applicable ROk by solving a simplified entailment problem:
executed(xs) |= βk. The algorithm may stop here, reporting a failure to indicate
that Dcomb is still inconsistent, if it has failed to address the encountered direct
interaction.

The second phase identifies and addresses indirect adverse interactions. It
starts by identifying applicable interaction operators (for an operator IOk this
translates to checking the entailment Dcomb |= αk). If there is no applicable
operator, then this means that there are no indirect interactions or they have
been already addressed, and the algorithm finds a model for Dcomb. This model
is equivalent to a solution in the CLP-based mitigation framework, and using its
Ipredicate component it is possible to construct a combined therapy for a patient.

This combined therapy highlights the clinical actions to be taken (executed and
dosage predicates) along with the order in which they should be carried out (prec
predicates), and includes the assumptions made about the patient’s state (value
predicates). Note that the combined therapy contains only these predicates that
have not been provided as part of Dpi, thus it is focused on future (suggested)
actions and possible (assumed) patient state.

On the other hand, if direct interactions exist (there is at least one IOk

applicable to Dcomb), the algorithm attempts to revise Dcomb using applicable
revision operators, where checking applicability of an operator ROk is formulated
as an entailment problem (Dcomb |= βk). In our previous research we assumed
that an interaction had to be addressed by a single applicable revision operator.
In this framework we relax this assumption and allow for more complex adverse
interactions that may need to be mitigated by multiple revision operators. There
is an additional explicit check if Dcomb has been revised to avoid indefinite loops if
there is no applicable ROk. If the revised Dcomb is consistent, then the algorithm
checks again for an applicable IOk, otherwise it fails. This loop is repeated until
there are no more applicable interaction operators.

The implementation of the mitigation algorithm involves a number of soft-
ware tools that were developed for FOL theories. In this research we are using
Prover9 [19] to check consistency of all theories and to execute the entailment
required for the identification and use of the operators. Moreover, we are using
a model finding technique implemented in Mace4 [6] that returns a model on
top of a theory that has been verified as a consistent one. The performance of
Prover9 was verified on a set of benchmark FOL problems and compared to
other solvers in [10]. The results show it was among two best performing solvers
and for most of the considered problems the proofs were generated in seconds
when running on a personal computer. The running times we observed in our
tests were comparable or even shorter, thus they are negligible with regards to
the patient management process. Moreover, the mitigation algorithm and its
implementation are not bound to Prover9 and Mace4, thus they can be easily
replaced by more efficient solvers, if performance becomes an issue.

5 Illustrative Example

In this section we illustrate our proposed FOL-based mitigation framework using
the simple clinical case also used in [19]. The purpose of using the same example
is to show how the methodology proposed here extends our earlier research.
According to this example, a patient, who is treated for a duodenal ulcer (DU),
experiences an episode of transient ischemic attack (TIA). AGs used in this
example are derived from the guidelines published by the National Institute for
Health and Clinical Excellence, UK (NICE) [9] and they have been simplified to
include only the relevant action and decision nodes.

Fig. 2. Actionable graph for DU (AGDU ).

5.1 Actionable Graphs

Figures 2 and 3 present AGs for DU and TIA simplified guidelines respectively.
In these figures the context nodes are indicated with circles, decision nodes are
indicated with diamonds, and action nodes with rectangles. The figures also
label constants associated with specific nodes and constants corresponding to
alternative choices – they are given in square brackets after node and choice
descriptions. For example, the HP constant is associated with the “H.pylori”
decision node (checking for the presence of helicobacter pylori). There are two
alternative choices at this decision node positive and negative. They are repre-
sented as P and N constants respectively.

5.2 Theories

The AGs are converted into the respective theories, DDU
cpg for DU and DTIA

cpg

for TIA, illustrated in Figs. 4 and 5. As can be seen, this representation cap-
tures precedence relationships and attaches semantics to each node. All paths
in the corresponding AG are described using a single sentence (a disjunction of
conjunctions, where each conjunction corresponds to a single path). Each path
contains formulas with the negated executed predicate to indicate these actions
are not executed for a given path.

Fig. 3. Actionable graph for TIA (AGTIA).

disease(DU).
node(HP ). node(ET ). node(ZES). node(PPI). node(UE). node(SC). node(RS).
directPrec(HP, ET ). directPrec(HP, ZES). directPrec(ET, PPI). directPrec(ZES, PPI).
directPrec(PPI, UE). directPrec(UE, SC). directPrec(UE, RS). directPrec(ZES, RS).
decision(HP ). decision(ZES). decision(UE).
action(ET ). action(PPI). action(SC). action(RS).
(value(HP, P ) ∧ executed(ET ) ∧ executed(PPI) ∧ value(UE, H) ∧ executed(SC)

∧ ¬executed(RS))
∨ (value(HP, P ) ∧ executed(ET ) ∧ executed(PPI) ∧ value(UE, NH) ∧ executed(RS)

∧ ¬executed(SC))
∨ (value(HP, N) ∧ value(ZES, N) ∧ executed(PPI) ∧ value(UE, H) ∧ executed(SC) ∧

¬executed(ET ) ∧ ¬executed(RS))
∨ (value(HP, N) ∧ value(ZES, N) ∧ executed(PPI) ∧ value(UE, NH) ∧ executed(RS)

∧ ¬executed(ET ) ∧ ¬executed(SC))
∨ (value(HP, N) ∧ value(ZES, P ) ∧ executed(RS) ∧

¬executed(ET ) ∧ ¬executed(PPI) ∧ ¬executed(SC)).

Fig. 4. The DDU
cpg theory representing the CPG for DU.

5.3 Operators

Interaction and revision operators associated with clinical scenarios discussed
below are given in Fig. 6 (for clarity only most relevant operations within revision
operators are presented). Their interpretation is as follows:

disease(TIA).
node(HG). node(FAST ). node(EC). node(NS). node(A). node(TST ). node(RST ).
node(PCS). node(D). node(NC).
directPrec(HG, FAST ). directPrec(HG, EC). directPrec(FAST, PCS).
directPrec(FAST, NS). directPrec(NS, A). directPrec(NS, TST ). directPrec(A, RST ).
directPrec(RST, PCS). directPrec(RST, D). directPrec(D, NC). directPrec(TST, NC).
decision(HG). decision(FAST ). decision(NS). decision(RST ).
action(EC). action(A). action(TST ). action(PCS). action(D). action(NC).
dosage(A, 300). dosage(D, 75).
(value(HG, N) ∧ value(FAST, N) ∧ executed(PCS)

∧ ¬executed(EC) ∧ ¬executed(A) ∧ ¬executed(TST ) ∧ ¬executed(D)
∧ ¬executed(NC))

∨ (value(HG, N) ∧ value(FAST, P ) ∧ value(NS, R) ∧ executed(A)
∧ value(RST, NG) ∧ executed(PCS)
∧ ¬executed(EC) ∧ ¬executed(TST ) ∧ ¬executed(D) ∧ ¬executed(NC))

∨ (value(HG, N) ∧ value(FAST, P ) ∧ value(NS, R) ∧ executed(A)
∧ value(RST, EL) ∧ executed(D) ∧ executed(NC)
∧ ¬executed(EC) ∧ ¬executed(TST ) ∧ ¬executed(PCS))

∨ (value(HG, N) ∧ value(FAST, P ) ∧ value(NS, NR) ∧ executed(TST ) ∧ executed(NC)
∧ ¬executed(EC) ∧ ¬executed(A) ∧ ¬executed(PCS) ∧ ¬executed(D))

∨ (value(HG, P ) ∧ executed(EC)
∧ ¬executed(A)) ∧ ¬executed(TST ) ∧ ¬executed(PCS) ∧ ¬executed(D)
∧ ¬executed(NC)).

Fig. 5. The DTIA
cpg theory representing the CPG for TIA.

Interaction operators:
IO1 =

〈
α1

〉

α1 = diagnosed(DU) ∧ executed(A) ∧ ¬executed(PPI)

Revision operators:
RO1 =

〈
β1, Op1

〉

β1 = diagnosed(DU) ∧ executed(A) ∧ ¬executed(PPI) ∧ ¬executed(D)

Op1 = {〈executed(A), executed(CL)〉}
RO2 =

〈
β2, Op2

〉

β2 = diagnosed(DU) ∧ executed(A) ∧ ¬executed(PPI) ∧ executed(D)

Op2 = {〈¬executed(PPI), executed(PPI)〉 , 〈dosage(A, x), dosage(A, x − 50〉}

Fig. 6. Interaction and revision operators.

– IO1 represents a drug-disease interaction (the increased risk of bleeding)
that occurs when a DU patient is given aspirin (A) without a proton-pump
inhibitor (PPI).

– RO1 is applicable to a patient diagnosed with DU who has been prescribed
aspirin (A) without a proton-pump inhibitor (PPI), and has not been pre-
scribed dipyridamole (D). In such case, the patient is taken off of aspirin and
prescribed clopidogrel (CL).

– RO2 is applicable toapatientdiagnosedwithDUwhohasbeenprescribedaspirin
(A)without a proton-pump inhibitor (PPI), and also has been prescribed dipyri-
damole (D). In suchcase, thepatient is prescribedaproton-pump inhibitor (PPI)
and dosage of aspirin (A) is reduced by 50 milligrams (mg).

5.4 Scenario 1: No Adverse Interactions

In this scenario we assume a patient suffering from DU who has tested positive for
H.pylori (HP) and is undergoing eradication therapy (ET), on presentation to the
emergency department with TIA symptoms has tested negative for hypoglycemia
(HG) and the result of FAST test (FAST) is negative. The theory Dpi describing
this patient is given in Fig. 7.

diagnosed(DU). value(HP, P ). executed(ET ).
diagnosed(TIA). value(HG, N). value(FAST, N).

Fig. 7. The Dpi describing the patient information in Scenario 1.

We create a theory Dcomb to describe this specific patient encounter, where
Dcpg are the union of DDU

cpg and DTIA
cpg discussed in Sect. 5.2.

The mitigation algorithm begins by applying theorem proving technique and
checking if Dcomb is consistent. Since the theory is consistent, the algorithm infers
that no direct interactions exist. At this stage the mitigation algorithm proceeds
to the second phase and checks for the existence of an indirect interaction. It
starts with IO1 by formulating the entailment problem Dcomb |= α1. Because
α1 is not entailed by Dcomb (i.e., there exists at least one model, where α1 is
not satisfied), there are no indirect interactions present in the theory and the
mitigation algorithm uses model finding techniques to find a model for the theory
Dcomb. One such model is found and used to create a combined therapy given in
Fig. 8 (for brevity we omitted the prec predicates).

executed(PPI). value(UE, H). executed(SC). executed(PCS).

Fig. 8. Combined therapy created for Scenario 1.

According to the combined therapy the patient should be prescribed a proton-
pump inhibitor (executed(PPI)) and since the result of the endoscopy (UE) is
not known (neither value(UE,H) nor value(UE,NH) is included in Dpi), the
combined therapy assumes a healed ulcer (value(UE,H)) and suggests self-
care (executed(SC)) for DU and a referral to a primary care specialist for TIA
(executed(PCS)). Such a combined therapy is returned by the mitigation algo-
rithm and presented to the physician along with the known patient state (Dpi).
The physician evaluates the therapy by checking the appropriateness of assump-
tions made, such as the assumption of a healed ulcer in this particular scenario.
If she deems some of these assumptions to be inappropriate, new patient infor-
mation needs to be collected and the mitigation algorithm needs to be invoked
again to generate a new combined therapy.

5.5 Scenario 2: Adverse Interactions Present

In this scenario we consider a patient suffering from DU, who has tested negative
for H.pylori (HP) and positive for Zollinger-Ellison syndrome (ZES), and who
on presentation to the emergency department with TIA symptoms has tested
negative for hypoglycemia (HG), passed FAST test, has had neurological symp-
toms (NS) resolved, and for whom the risk of stroke (RST) has been evaluated
as elevated. The theory Dpi describing this patient is given in Fig. 9.

diagnosed(DU). value(HP, N). value(ZES, P ).
diagnosed(TIA). value(HG, N). value(FAST, P ). value(NS, R). value(RST, EL).

Fig. 9. The Dpi describing the patient information in Scenario 2.

Similarly to the previous scenario, Dcomb is consistent and as such no direct
interactions exist. To check for the existence of an indirect interaction we consider
IO1 and formulate the entailment problem Dcomb |= α1. This time α1 is entailed
by Dcomb (it is satisfied by each model of Dcomb) indicating that an indirect
interaction exists.

Following the steps of the mitigation algorithm, we resolve an indirect interac-
tion by selecting a relevant revision operator to revise Dcpg. A relevant operator
is found by iterating over available revision operators and formulating the entail-
ment problem Dcpg |= βk for each revision operator ROk. In this scenario, for
RO1 β1 is not entailed by Dcomb as there exists at least one model that does not
satisfy β1. This indicates that RO1 is not a relevant revision operator. Next, the
algorithm considers RO2 and formulates the entailment problem Dcomb |= β2.
Now β2 is entailed by Dcomb and RO2 is considered a relevant revision operator.

(PPI) (RS).

executed(A). dosage(A, 250) (D) (D, 75). executed(NC).

Fig. 10. Combined therapy created for Scenario 2 (underlined entries have been intro-
duced by the revision operator).

The algorithm revises Dcomb by modifying Dcpg according to the operations
Op2 defined in RO2. These operations introduce a proton pump inhibitor (in fact
¬executed (PPI) is replaced by executed(PPI) to avoid direct interaction) and
reduce the dosage of aspirin by 50 mg to 250 mg (replacing dosage(A, 300) with
dosage(A, 250)). After making these revisions, the mitigation algorithm checks
if the revised Dcomb is consistent. Since it is, the algorithm finds a model for the
revised Dcomb that includes the modified Dcpg. This model is used to derive the
combined therapy given in Fig. 10 (again the prec predicates are excluded for
brevity).

According to the combined therapy, the patient is prescribed PPI (executed
(PPI)) and referred to a specialist for DU (executed(RS)). Also the therapy
prescribes aspirin (executed(A)) with the dosage adjusted to 250 mg (dosage(A,
250)), prescribes dipyridamole (executed(D)) with the dosage set to 75 mg
(dosage(D, 75)), and schedules an outpatient neurological consult for TIA (exe-
cuted(NC)). As in the previous scenario, such combined therapy is presented for
evaluation to the physician who may invoke the algorithm again once additional
patient information becomes available.

6 Conclusions

We believe that FOL allows for a more flexible representation by including pred-
icates to represent properties of domain objects, temporal relationships, and
flexibly quantified sentences. In this paper we presented how using FOL theories
allows us to augment the expressiveness of representation in order to capture
intrinsic characteristics of the CPGs and combined therapies, and thus provides
for a more complete mitigation framework. Using a simple clinical example we
demonstrated the semantic interpretability of the models and combined ther-
apies. In our earlier CLP-based framework we had to manually interpret the
solutions, distinguishing between action and decision steps, and constructing
temporal relationships to impose order in which steps should be taken. The new
framework discussed here addresses all these shortcomings.

Presented new framework allows us to deal with such “hard” issues associated
with CPGs as, for example, loops. This improved expressiveness comes at the
cost of limited comprehensibility by non-experts. However, considering that we
envisage the proposed framework to be embedded within a larger clinical decision
support system that will present results of mitigation in a user-friendly way, a
modeling complexity should not be an issue because actual model will not be
seen/presented to a clinician. Only development of the operators will require
direct involvement of a clinician, and this process will be guided by a knowledge
transfer specialist.

For future research, we are working on a different representation of paths in
Ddi

cpg, so disjunctions of conjunctions can be avoided, and on more sophisticated
search methods employed by the mitigation algorithm to identify suitable revi-
sion operators. Considering that the ultimate goal of our research is to develop
a generalized framework of mitigation, we are also studying different clinical sit-
uations involving comorbid patients to extract the full set of properties of CPGs
that hold across mitigation scenarios.

